
[bookmark: _Toc122434485][bookmark: _Toc108004478]Overview
This document lists all the essential changes needed to correct problems in the TTCN implementation of NB-IoT NTN test case 22.3.2.7a which is part of the NBIOT test suite in iwd-TTCN3-B2022-09_D23wk24.
With these changes applied the test case can be demonstrated to run with one or more UEs (see section 4). Execution log file(s) is (are) provided as evidence. 
Contact:	Parikshit Bhise
	parikshit.bhise@rohde-schwarz.com 
	
0. Verification Test Summary 
Test Case: 	22_3_2_7a_NBIOT
ATS Version:	iwd-TTCN3-B2022-09_D23wk24
System Simulator used:	R&S® CMW 500 Protocol Conformance Test platform
UE used:	MediaTek MTK 6825, Qualcomm 9205S
Verification Status:	PASS



[bookmark: _Toc493251607]Corrections required for test case 22.3.2.7a
f_TC_22_3_2_7a_NBIOT
	Function name
	f_TC_22_3_2_7a_NBIOT

	Reason for change
	After switch on, the function f_UT_AutomaticPLMN_Select_WithSwitchOnOff calls switch off procedure, but there is no switch on procedure to follow up.

	Summary of change
	Removed the function and replaced it with f_UT_AutomaticPLMN_Select to avoid the additional switch off call

	TTCN module
	NBIOT_RLC_NTN

	MCC160 Comment
	



Before change
	function f_TC_22_3_2_7a_NBIOT() runs on NBIOT_PTC
  { /* NB-IoT / NTN / AM RLC / Receiver status triggers / extended t-Reordering configured */
    var RLC_SS_State v_RLC_Rec;
    var NB_RLC_CountsInfoList_Type v_RLC_CountsInfoList;
    var template (value) DL_CCCH_Message_NB v_RrcConnSetup;
    var T_PollRetransmit_NB_r13 v_T_PollRetransmit_NB_r13;
    var RLC_Config_NB_v1430.t_Reordering_r14 v_T_Reordering_r14;
    var RLC_Config_NB_v1700.t_ReorderingExt_r17.setup v_T_Reordering_r17;
    var UL_AM_RLC_NB_r13.maxRetxThreshold_r13 v_MaxRetxThreshold_r13;
    var template (value) SRB_ToAddModList_NB_r13 v_SRB_ToAddModList_NB_r13;
    var integer v_HARQ_ProcessNumber := 0;
    
    //System information combination 8 as defined in TS 36.508 [18] clause 8.1.4.3.1.1 is used.
    f_NBIOT_Init(c8);
    
    //Create and configure all cells
    f_NBIOT_CellConfig_Def(nbiot_Cell1, CONTROL_PLANE);
      
    //The HARQ process 0 is used for DL and UL transmissions as default.
    f_NBIOT_SS_CommonCellConfig(nbiot_Cell1, cas_NB_CcchDcchDtchDL_Config_REQ(nbiot_Cell1, cs_TimingInfo_Now, -, cs_NB_DciDlInfo_Auto(v_HARQ_ProcessNumber)));
    
    //RLC global initialisation
    f_InitRLC_Record(v_RLC_Rec, AM_Mode);
      
    //Initialise acc. Table 22.3.2.7a.3.3-1
    v_T_PollRetransmit_NB_r13 := ms4000;
    v_RLC_Rec.t_PollRetransmit_r13 := 4.0; // t_PollRetransmit_r13 set to 4000 ms
      
    v_MaxRetxThreshold_r13 := t2;
    v_RLC_Rec.maxRetxThreshold_r13 := 2; // maxRetxThreshold_r13 set to 2 (=t2)
            
    v_RLC_Rec.enableStatusReportSN_Gap_r13 := true; // enableStatusReportSN-Gap-r13 set to true
    
    v_T_Reordering_r14 := ms1600_v1310;
    v_RLC_Rec.t_Reordering_r14 := 1.6; // t_Reordering_r14 set to 1600 ms
      
    v_T_Reordering_r17 := ms3200;
    v_RLC_Rec.t_Reordering_r17 :=3.2; //t_Reordering_r17 set to 3200 ms
    
    //UE RLC settings are as initialized
    v_SRB_ToAddModList_NB_r13 := {cs_SRB_ToAddMod_NB_RLC_Config (cs_UL_AM_RLC_NB_r13 (v_T_PollRetransmit_NB_r13, v_MaxRetxThreshold_r13),
                                                                 cs_DL_AM_RLC_NB_r13_GapReport,
                                                                 cs_RLC_Config_NB_v1430(v_T_Reordering_r14))};
    v_SRB_ToAddModList_NB_r13[0].rlc_Config_v1700.t_ReorderingExt_r17.setup := v_T_Reordering_r17; //acc to Table 22.3.2.7a.3.3-1

    // RRCConnectionSetup-NB with twoHARQ-ProcessesConfig-r14 configured
    v_RrcConnSetup := f_NBIOT_508_RRCConnectionSetup(-, v_SRB_ToAddModList_NB_r13, -, -, true_);
    f_NBIOT_SS_ConfigRachProcedureMsg4(nbiot_Cell1, v_RrcConnSetup);
        
    f_NBIOT_SetCellPower(nbiot_Cell1, tsc_ServingCellRS_EPRE);
      
    //The UE is in state 2B-NB according to TS 36.508 [18] clause 8.1.5.2B with test loop mode G closed
    
    // Switch on UE
    f_UT_SwitchOnUE(UT, false);
      
    //The UE is in Automatic PLMN selection mode
    f_UT_AutomaticPLMN_Select_WithSwitchOnOff (UT);
      
    //The preconfigured UE location is defined in TS 36.508 [18] Clause 4.13.
    f_UT_ConfigureNTNPosition(UT);
      
    // Now continue with the procedure
    f_NBIOT_InitialRegistration(nbiot_Cell1, CONTROL_PLANE, PREAMBLE, STATE2A_NB_TESTLOOP_ModeG);
    



After change
	function f_TC_22_3_2_7a_NBIOT() runs on NBIOT_PTC
  { /* NB-IoT / NTN / AM RLC / Receiver status triggers / extended t-Reordering configured */
    var RLC_SS_State v_RLC_Rec;
    var NB_RLC_CountsInfoList_Type v_RLC_CountsInfoList;
    var template (value) DL_CCCH_Message_NB v_RrcConnSetup;
    var T_PollRetransmit_NB_r13 v_T_PollRetransmit_NB_r13;
    var RLC_Config_NB_v1430.t_Reordering_r14 v_T_Reordering_r14;
    var RLC_Config_NB_v1700.t_ReorderingExt_r17.setup v_T_Reordering_r17;
    var UL_AM_RLC_NB_r13.maxRetxThreshold_r13 v_MaxRetxThreshold_r13;
    var template (value) SRB_ToAddModList_NB_r13 v_SRB_ToAddModList_NB_r13;
    var integer v_HARQ_ProcessNumber := 0;
    
    //System information combination 8 as defined in TS 36.508 [18] clause 8.1.4.3.1.1 is used.
    f_NBIOT_Init(c8);
    
    //Create and configure all cells
    f_NBIOT_CellConfig_Def(nbiot_Cell1, CONTROL_PLANE);
      
    //The HARQ process 0 is used for DL and UL transmissions as default.
    f_NBIOT_SS_CommonCellConfig(nbiot_Cell1, cas_NB_CcchDcchDtchDL_Config_REQ(nbiot_Cell1, cs_TimingInfo_Now, -, cs_NB_DciDlInfo_Auto(v_HARQ_ProcessNumber)));
    
    //RLC global initialisation
    f_InitRLC_Record(v_RLC_Rec, AM_Mode);
      
    //Initialise acc. Table 22.3.2.7a.3.3-1
    v_T_PollRetransmit_NB_r13 := ms4000;
    v_RLC_Rec.t_PollRetransmit_r13 := 4.0; // t_PollRetransmit_r13 set to 4000 ms
      
    v_MaxRetxThreshold_r13 := t2;
    v_RLC_Rec.maxRetxThreshold_r13 := 2; // maxRetxThreshold_r13 set to 2 (=t2)
            
    v_RLC_Rec.enableStatusReportSN_Gap_r13 := true; // enableStatusReportSN-Gap-r13 set to true
    
    v_T_Reordering_r14 := ms1600_v1310;
    v_RLC_Rec.t_Reordering_r14 := 1.6; // t_Reordering_r14 set to 1600 ms
      
    v_T_Reordering_r17 := ms3200;
    v_RLC_Rec.t_Reordering_r17 :=3.2; //t_Reordering_r17 set to 3200 ms
    
    //UE RLC settings are as initialized
    v_SRB_ToAddModList_NB_r13 := {cs_SRB_ToAddMod_NB_RLC_Config (cs_UL_AM_RLC_NB_r13 (v_T_PollRetransmit_NB_r13, v_MaxRetxThreshold_r13),
                                                                 cs_DL_AM_RLC_NB_r13_GapReport,
                                                                 cs_RLC_Config_NB_v1430(v_T_Reordering_r14))};
    v_SRB_ToAddModList_NB_r13[0].rlc_Config_v1700.t_ReorderingExt_r17.setup := v_T_Reordering_r17; //acc to Table 22.3.2.7a.3.3-1

    // RRCConnectionSetup-NB with twoHARQ-ProcessesConfig-r14 configured
    v_RrcConnSetup := f_NBIOT_508_RRCConnectionSetup(-, v_SRB_ToAddModList_NB_r13, -, -, true_);
    f_NBIOT_SS_ConfigRachProcedureMsg4(nbiot_Cell1, v_RrcConnSetup);
        
    f_NBIOT_SetCellPower(nbiot_Cell1, tsc_ServingCellRS_EPRE);
      
    //The UE is in state 2B-NB according to TS 36.508 [18] clause 8.1.5.2B with test loop mode G closed
    
    // Switch on UE
    f_UT_SwitchOnUE(UT, false);
      
    //The UE is in Automatic PLMN selection mode
    // REMOVED f_UT_AutomaticPLMN_Select_WithSwitchOnOff (UT);
    f_UT_AutomaticPLMN_Select (UT); //(similar to other test cases
    //The preconfigured UE location is defined in TS 36.508 [18] Clause 4.13.
    f_UT_ConfigureNTNPosition(UT);
      
    // Now continue with the procedure
    f_NBIOT_InitialRegistration(nbiot_Cell1, CONTROL_PLANE, PREAMBLE, STATE2A_NB_TESTLOOP_ModeG);
    



fl_TC_22_3_2_7a_TestBody
	Function name
	fl_TC_22_3_2_7a_TestBody

	Reason for change
	1. In order to transmit the RLC PDUs the SDUs are not generated and results in runtime error
2. When the 3 RLC PDUs are transmitted, the SS cannot honour the current activation time specified as the next search space cycle is 64msec, and this does not consider the Koffset into account, since the Koffset is 550msec, updated the timing to use 10* search space (640msec) as p_Offset
3. After Step 6, the DL data cannot be sent on the next PDCCH cycle due to Koffset, therefore updated to use 10* speech space (640msec) as p_Offset

4. When checking for the timer duration , the test case needs to take into account the Koffset value
5. Since the test procedure triggers out of order and UE generated a status report with NACK PDU, the current prose does not transmit the NACK PDU, and this results in test case failure at postamble, In TTCN, in order to perform postamble procedure the RLC count is updated to the SS and SRB3 is moved from test mode to normal mode). SS RLC is now aware of the NACK PDU and results in TC failure. In order to avoid this additional steps are added for UE to loopback the received data is necessary. 


	Summary of change
	1. Added f_NBIOT_Generate_RLC_SDUs
2. v_TimingInfo1 and v_TimingInfo2 is updated to use p_Offset  value as 11 and 21 and step 2 grant will be triggered in 31
3. Step 8 data is started with p_Offset as 10 and grant is sent with p_Offset as 11
4. Introduced new function to take into account the Koffset for max tolerance 
5. Introduced Steps 10-17 to transmit the NACK PDU and receive all loopback data before proceeding into postamble

Changes proposed here requires an update to 36.523-3

	TTCN module
	NBIOT_RLC_NTN

	MCC160 Comment
	



Before change
	function fl_TC_22_3_2_7a_TestBody(inout RLC_SS_State p_RLC_Rec) runs on NBIOT_PTC
  {
    var integer v_Duration;
    var template (value) TimingInfo_Type v_TimingInfo;
    var template (value) TimingInfo_Type v_TimingInfo1;
    var template (value) TimingInfo_Type v_TimingInfo2;
    var template (present) RLC_Status_NACK_List_Type v_NACK_List;
    var template (present) RLC_Status_Padding_Type v_Padding_Any := ?;
    var template (value) NB_DciUlInfo_Type v_DciUlInfo;
    var integer v_VTS_Offset := p_RLC_Rec.AM_VTS;
    var SubFrameTiming_Type v_TimingStart;
    var SubFrameTiming_Type v_SubFrameTimingTa;
    var SubFrameTiming_Type v_SubFrameTimingTb;
    
    
    //The SS does not respond to PRACH preambles transmitted by UE for Uplink transmission, but instead allocates the UL C-RNTI grant on NPDCCH when specified in the test sequence.
    //Note: In steps 1 to 8 the size of each RLC UL SDU #1 - #3 is PRBS of 288 bits (36 octets). The size of the RLC SDUs used in downlink is L.

    //@siclog "Step 1" siclog@
    //The SS transmits 3 AMD PDUs with SN=0, 1, 2. The SS sets the P field of all the AMD PDUs to 0.
    v_TimingStart := f_NBIOT_GetNextSearchSpace(nbiot_Cell1);
    v_TimingInfo := cs_TimingInfo_SubFrameFull(f_NBIOT_IncrementSearchSpace(v_TimingStart, 1.0));
    v_TimingInfo1 := cs_TimingInfo_SubFrameFull(f_NBIOT_IncrementSearchSpace(v_TimingStart, 2.0));
    v_TimingInfo2 := cs_TimingInfo_SubFrameFull(f_NBIOT_IncrementSearchSpace(v_TimingStart, 3.0));
    f_NB_TxAMD_SDU(p_RLC_Rec, v_TimingInfo, tsc_P_NoPoll, tsc_RLC_SDU1);
    f_NB_TxAMD_SDU(p_RLC_Rec, v_TimingInfo1, tsc_P_NoPoll, tsc_RLC_SDU2);
    f_NB_TxAMD_SDU(p_RLC_Rec, v_TimingInfo2, tsc_P_NoPoll, tsc_RLC_SDU3);

    //@siclog "Step 2" siclog@
    //In the search space of the 4th NPDCCH period after the first transmission at step 1
    //the SS schedules 3 consecutive UL grants with a time spacing of 3 NPDCCH cycles of size 328 bits.
    v_TimingInfo := cs_TimingInfo_SubFrameFull(f_NBIOT_IncrementSearchSpace(v_TimingStart, 5.0));
    v_DciUlInfo := cs_NB_DciInfo_CcchDcchDtchUL (1,10);  // UL Grant of 328 bits
    f_NBIOT_ULGrantTransmission(nbiot_Cell1, cs_NB_UL_GrantScheduling_Start(cs_NB_PeriodicGrant(v_DciUlInfo, 3, 3)), v_TimingInfo);

    //@siclog "Step 3" siclog@
    //The UE transmits RLC UL SDU#1.
    f_NB_RxAMD_SDU(p_RLC_Rec, tsc_P_NoPoll, tsc_RLC_SDU1);
   
    //@siclog "Step 4" siclog@
    //The UE transmits RLC UL SDU#2.
    f_NB_RxAMD_SDU(p_RLC_Rec, tsc_P_NoPoll, tsc_RLC_SDU2);

    //@siclog "Step 5" siclog@
    //The UE transmits RLC UL SDU#3.
    f_NB_RxAMD_SDU(p_RLC_Rec, tsc_P_Poll, tsc_RLC_SDU3);

    //@siclog "Step 6" siclog@
    //The SS transmits a STATUS PDU
    f_NB_TxSTATUS_PDU (p_RLC_Rec);

    //@siclog "Step 7" siclog@
    //The SS starts the UL periodic grant transmission of size 40 bits.
    v_TimingStart := f_NBIOT_GetNextSearchSpace(nbiot_Cell1);
    v_TimingInfo := cs_TimingInfo_SubFrameFull(f_NBIOT_IncrementSearchSpace(v_TimingStart, 1.0));
    v_DciUlInfo := cs_NB_DciInfo_CcchDcchDtchUL (0,3);
    f_NBIOT_ULGrantTransmission(nbiot_Cell1, cs_NB_UL_GrantScheduling_Start(cs_NB_ContinuousGrant(v_DciUlInfo)), v_TimingInfo);

    //@siclog "Step 8" siclog@
    //The SS transmits 1 AMD PDUs with SN=4 and P=0.
    //Record time TA.
    v_SubFrameTimingTa := f_NBIOT_IncrementSearchSpace(v_TimingStart, 2.0);
    v_TimingInfo := cs_TimingInfo_SubFrameFull(v_SubFrameTimingTa);
    p_RLC_Rec.AM_VTS := v_VTS_Offset + 4;
    f_NB_TxAMD_SDU(p_RLC_Rec, v_TimingInfo, tsc_P_NoPoll, tsc_RLC_SDU5);

    //@siclog "Step 9" siclog@
    //Check 1: Does the UE transmit a Status Report with NACK_SN=3 and ACK_SN=5?
    //Record time TB
    // Check 2: (TB - TA) = t-Reordering?
    p_RLC_Rec.AM_VTA := v_VTS_Offset + 5;
    v_NACK_List := cr_NACK_List(v_VTS_Offset + 3, tsc_E2_NoSO_Follows, omit);
    f_NB_RxSTATUS_PDU_NACK(p_RLC_Rec, v_NACK_List, v_Padding_Any);
    f_NBIOT_PreliminaryPass(__FILE__, __LINE__, "Step 9 Check 1");
    v_SubFrameTimingTb := p_RLC_Rec.TimeStampLastReceivedPDU;
    v_Duration := f_SubFrameTiming_Duration( v_SubFrameTimingTa, v_SubFrameTimingTb);
    if ( f_NBIOT_SubFrameTimingCheckDuration( l2Timer, v_Duration, float2int(p_RLC_Rec.t_Reordering_r17 * 1000.0) ) )
    {
      f_NBIOT_PreliminaryPass(__FILE__, __LINE__, "Step 9 Check 2");
    }
    else
    {
      f_NBIOT_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 9 Check 2");
    };
    
  }
    
}



After change
	  import from EUTRA_NB_CommonL2 all;
…
…
…


function fl_TC_22_3_2_7a_TestBody(inout RLC_SS_State p_RLC_Rec) runs on NBIOT_PTC
  {
    var integer v_RLC_DL_SDU_Length;
    var integer v_RLC_UL_SDU_Length := 36;
    var integer v_L;  
    var integer v_Duration;
    var template (value) TimingInfo_Type v_TimingInfo;
    var template (value) TimingInfo_Type v_TimingInfo1;
    var template (value) TimingInfo_Type v_TimingInfo2;
    var template (present) RLC_Status_NACK_List_Type v_NACK_List;
    var template (present) RLC_Status_Padding_Type v_Padding_Any := ?;
    var template (value) NB_DciUlInfo_Type v_DciUlInfo;
    var integer v_VTS_Offset := p_RLC_Rec.AM_VTS;
    var SubFrameTiming_Type v_TimingStart;
    var SubFrameTiming_Type v_SubFrameTimingTa;
    var SubFrameTiming_Type v_SubFrameTimingTb;
    
    
    //The SS does not respond to PRACH preambles transmitted by UE for Uplink transmission, but instead allocates the UL C-RNTI grant on NPDCCH when specified in the test sequence.
    //Note: In steps 1 to 8 the size of each RLC UL SDU #1 - #3 is PRBS of 288 bits (36 octets). The size of the RLC SDUs used in downlink is L.
    // generate SDUs with PRBS size 36 octets
    
    v_RLC_DL_SDU_Length := f_NBIOT_Generate_RLC_SDUs (p_RLC_Rec, 6, v_RLC_UL_SDU_Length, 0);
    v_L := v_RLC_DL_SDU_Length - v_RLC_UL_SDU_Length; // RLC DL SDU length - RLC UL SDU length, Length difference is the NAS and RRC  overhead
    //@siclog "Step 1" siclog@
    //The SS transmits 3 AMD PDUs with SN=0, 1, 2. The SS sets the P field of all the AMD PDUs to 0.
    v_TimingStart := f_NBIOT_GetNextSearchSpace(nbiot_Cell1);
    v_TimingInfo := cs_TimingInfo_SubFrameFull(f_NBIOT_IncrementSearchSpace(v_TimingStart, 1.0));
    v_TimingInfo1 := cs_TimingInfo_SubFrameFull(f_NBIOT_IncrementSearchSpace(v_TimingStart, 11.0));
    v_TimingInfo2 := cs_TimingInfo_SubFrameFull(f_NBIOT_IncrementSearchSpace(v_TimingStart, 21.0));
    f_NB_TxAMD_SDU(p_RLC_Rec, v_TimingInfo, tsc_P_NoPoll, tsc_RLC_SDU1);
    f_NB_TxAMD_SDU(p_RLC_Rec, v_TimingInfo1, tsc_P_NoPoll, tsc_RLC_SDU2);
    f_NB_TxAMD_SDU(p_RLC_Rec, v_TimingInfo2, tsc_P_NoPoll, tsc_RLC_SDU3);

    //@siclog "Step 2" siclog@
    //In the search space of the 4th NPDCCH period after the first transmission at step 1
    //the SS schedules 3 consecutive UL grants with a time spacing of 3 NPDCCH cycles of size 328 bits.
    v_TimingInfo := cs_TimingInfo_SubFrameFull(f_NBIOT_IncrementSearchSpace(v_TimingStart, 31.0));
    v_DciUlInfo := cs_NB_DciInfo_CcchDcchDtchUL (1,10);  // UL Grant of 328 bits
    f_NBIOT_ULGrantTransmission(nbiot_Cell1, cs_NB_UL_GrantScheduling_Start(cs_NB_PeriodicGrant(v_DciUlInfo, 3, 3)), v_TimingInfo);

    //@siclog "Step 3" siclog@
    //The UE transmits RLC UL SDU#1.
    f_NB_RxAMD_SDU(p_RLC_Rec, tsc_P_NoPoll, tsc_RLC_SDU1);
   
    //@siclog "Step 4" siclog@
    //The UE transmits RLC UL SDU#2.
    f_NB_RxAMD_SDU(p_RLC_Rec, tsc_P_NoPoll, tsc_RLC_SDU2);

    //@siclog "Step 5" siclog@
    //The UE transmits RLC UL SDU#3.
    f_NB_RxAMD_SDU(p_RLC_Rec, tsc_P_Poll, tsc_RLC_SDU3);

    //@siclog "Step 6" siclog@
    //The SS transmits a STATUS PDU
    f_NB_TxSTATUS_PDU (p_RLC_Rec);

    //@siclog "Step 7" siclog@
    //The SS starts the UL periodic grant transmission of size 40 bits.
    v_TimingStart := f_NBIOT_GetNextSearchSpace(nbiot_Cell1);
    v_TimingInfo := cs_TimingInfo_SubFrameFull(f_NBIOT_IncrementSearchSpace(v_TimingStart, 11.0));
    v_DciUlInfo := cs_NB_DciInfo_CcchDcchDtchUL (0,3);
    f_NBIOT_ULGrantTransmission(nbiot_Cell1, cs_NB_UL_GrantScheduling_Start(cs_NB_ContinuousGrant(v_DciUlInfo)), v_TimingInfo);

    //@siclog "Step 8" siclog@
    //The SS transmits 1 AMD PDUs with SN=4 and P=0.
    //Record time TA.
    v_SubFrameTimingTa := f_NBIOT_IncrementSearchSpace(v_TimingStart, 10.0); 
    v_TimingInfo := cs_TimingInfo_SubFrameFull(v_SubFrameTimingTa);
    p_RLC_Rec.AM_VTS := v_VTS_Offset + 4;
    f_NB_TxAMD_SDU(p_RLC_Rec, v_TimingInfo, tsc_P_NoPoll, tsc_RLC_SDU5);

    //@siclog "Step 9" siclog@
    //Check 1: Does the UE transmit a Status Report with NACK_SN=3 and ACK_SN=5?
    //Record time TB
    // Check 2: (TB - TA) = t-Reordering?
    p_RLC_Rec.AM_VTA := v_VTS_Offset + 5;
    v_NACK_List := cr_NACK_List(v_VTS_Offset + 3, tsc_E2_NoSO_Follows, omit);
    f_NB_RxSTATUS_PDU_NACK(p_RLC_Rec, v_NACK_List, v_Padding_Any);
    f_NBIOT_PreliminaryPass(__FILE__, __LINE__, "Step 9 Check 1");
    v_SubFrameTimingTb := p_RLC_Rec.TimeStampLastReceivedPDU;
    v_Duration := f_SubFrameTiming_Duration( v_SubFrameTimingTa, v_SubFrameTimingTb);
    if ( f_NBIOT_NTN_SubFrameTimingCheckDuration( l2Timer, v_Duration, float2int(p_RLC_Rec.t_Reordering_r17 * 1000.0) ) ) 
    {
      f_NBIOT_PreliminaryPass(__FILE__, __LINE__, "Step 9 Check 2");
    }
    else
    {
      f_NBIOT_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 9 Check 2");
    };
      
    
    //@siclog "Step 10" siclog@
    //SS stops to allocate any uplink grant.
    f_NBIOT_ULGrantTransmission(nbiot_Cell1, cs_NB_UL_GrantScheduling_Stop);
    
    v_TimingStart := f_NBIOT_GetNextSearchSpace(nbiot_Cell1);
    v_TimingInfo := cs_TimingInfo_SubFrameFull(f_NBIOT_IncrementSearchSpace(v_TimingStart, 1.0));
    v_TimingInfo1 := cs_TimingInfo_SubFrameFull(f_NBIOT_IncrementSearchSpace(v_TimingStart, 11.0));
    v_TimingInfo2 := cs_TimingInfo_SubFrameFull(f_NBIOT_IncrementSearchSpace(v_TimingStart, 21.0)); 
    
    //@siclog "Step 11" siclog@
    
    p_RLC_Rec.AM_VTS := v_VTS_Offset + 3;
    f_NB_TxAMD_SDU(p_RLC_Rec, v_TimingInfo, tsc_P_Poll, tsc_RLC_SDU4);

    //@siclog "Step 12" siclog@
    v_DciUlInfo := cs_NB_DciInfo_CcchDcchDtchUL (0,3);
    f_NBIOT_ULGrantTransmission(nbiot_Cell1, cs_NB_UL_GrantScheduling_Start(cs_NB_SingleGrant(v_DciUlInfo)), v_TimingInfo1);

    //@siclog "Step 14" siclog@ 
    //
    //Executed before step 13 to allow proper scheduling of UL grants !
    v_DciUlInfo := cs_NB_DciInfo_CcchDcchDtchUL (1,10);  // UL Grant of 328 bits
    f_NBIOT_ULGrantTransmission(nbiot_Cell1, cs_NB_UL_GrantScheduling_Start(cs_NB_PeriodicGrant(v_DciUlInfo, 2)), v_TimingInfo2);

    //@siclog "Step 13" siclog@
    //Check: Does the UE transmit a Status Report with no NACK_SN and ACK_SN = 5?
    p_RLC_Rec.AM_VTA := v_VTS_Offset + 5;
    f_NBNTN_RxSTATUS_PDU_ACK(p_RLC_Rec);
    f_NBIOT_PreliminaryPass(__FILE__, __LINE__, "Step 13");

    //@siclog "Step 15" siclog@
    //The UE transmits RLC UL SDU#4.
    f_NB_RxAMD_SDU(p_RLC_Rec, tsc_P_NoPoll, tsc_RLC_SDU4);

    //@siclog "Step 16" siclog@
    //The UE transmits RLC UL SDU#5.
    f_NB_RxAMD_SDU(p_RLC_Rec, tsc_P_Poll, tsc_RLC_SDU5);

    //@siclog "Step 17" siclog@
    //The SS transmits a STATUS PDU
    f_NB_TxSTATUS_PDU (p_RLC_Rec);
    p_RLC_Rec.AM_VTS := v_VTS_Offset + 5
    
  }
    
}




f_NBNTN_RxSTATUS_PDU_ACK
	Function name
	f_NBNTN_RxSTATUS_PDU_ACK

	Reason for change
	New function introduced to handle RxSTATUS_PDU_ACK, see change 2.2.5

	Summary of change
	Introduced new function (same as f_NB_RxSTATUS_PDU_ACK)


	TTCN module
	NBIOT_RLC_NTN

	MCC160 Comment
	




New function
	function f_NBNTN_RxSTATUS_PDU_ACK(RLC_SS_State p_RLC_Rec)
    runs on NBIOT_PTC
  {
    L2DATA.receive(car_NB_SRB1bis_RLCStatusInd(nbiot_Cell1,
                                               cr_RLC_AM_StatusPDU_ACK_SN(p_RLC_Rec.AM_VTA)));
  }  




f_NBIOT_NTN_SubFrameTimingCheckDuration
	Function name
	f_NBIOT_NTN_SubFrameTimingCheckDuration

	Reason for change
	New function introduced to handle SubframeTiming check and taking into account Koffset (see 2.2.4 change)

	Summary of change
	Introduced new function f_NBIOT_NTN_SubFrameTimingCheckDuration


	TTCN module
	NBIOT_RLC_NTN

	MCC160 Comment
	




New function
	function f_NBIOT_NTN_SubFrameTimingCheckDuration(ProtocolTimer_Type p_ProtocolTimer,
                                               integer p_Duration,
                                               integer p_ExpectedDuration,
                                               integer p_Koffset := 550) return boolean
  {
   
    var float v_ToleranceFloat := fl_NBIOT_TimerTolerance(p_ProtocolTimer, int2float(p_ExpectedDuration)/1000.0);
    var integer v_ToleranceInt := float2int(v_ToleranceFloat * 1000.0);
    var integer v_Min := p_ExpectedDuration - v_ToleranceInt;
    var integer v_Max := p_ExpectedDuration + v_ToleranceInt+ 2*p_Koffset;
     

    return ((v_Min <= p_Duration) and (p_Duration <= v_Max));
  }




[bookmark: _Toc325725665][bookmark: _Toc295288970][bookmark: _Toc122434493][bookmark: _Toc378008218][bookmark: _Toc493251612]Branches executed in test case 22.3.2.7a
The test case implementation was executed with Integrity and ciphering enabled (eea1 and eia1) using NTN bands 255 and 256.
[bookmark: _Toc325725666][bookmark: _Toc295288971][bookmark: _Toc122434494][bookmark: _Toc378008219][bookmark: _Toc493251613]Execution Log Files
Qualcomm 9205S
The Qualcomm 9205S device passed this test case on Rohde & Schwarz CMW500. The documentation below is enclosed as evidence of the successful test case run [1]:
[bookmark: OLE_LINK4][bookmark: OLE_LINK5][bookmark: OLE_LINK6]Test Case Execution log file: 
tc_22_3_2_7a_QC.log 
(Note: PICS/PIXIT settings are captured at the beginning of the TLI log)
[bookmark: _Toc493251615]MediaTek 6825
The MediaTek 6825 device passed this test case on Rohde & Schwarz CMW500. The documentation below is enclosed as evidence of the successful test case run [1]:
[bookmark: OLE_LINK1][bookmark: OLE_LINK2][bookmark: OLE_LINK3]Test Case Execution log file: 
tc_22_3_2_7a_MTK.log 
(Note: PICS/PIXIT settings are captured at the beginning of the TLI log)
n the log file the complete test case execution as well as the PICS/PIXIT parameter settings can be seen. All TLI events are presented and message contents are fully decoded and can be verified. Preliminary verdicts and the final test case verdict can be seen in the log file.
[bookmark: _Toc325725668][bookmark: _Toc295288973][bookmark: _Toc122434496][bookmark: _Toc378008221][bookmark: _Toc493251616]References
	[1]
	R5s230536:   Supporting information for addition of NB-IOT NTN testcase 22.3.2.7a.

